
Repetisjon: Abstraksjonsbarrierer

en-prosedyre-som-bruker-mengder

element-of-set?, adjoin-set, intersection-set, union-set

make-tree, entry, left-branch, right-branch

cons, car, cdr, null?, pair?

lambda, eller en annen par-implementasjon?

(En slags) Repetisjon: Kode som trær

I Ved evaluering oversettes kildekoden i et språk først til et syntakstre.
I Vi har sett at lister kan representere trær i Lisp.
I I Lisp uttrykkes også kode some lister.
I Lisp-kode uttrykker dermed syntaks-trær direkte!

(+ (- 100 60)
(/ 10

(+ 2 1 2)))

+

-

100 60

/

10 +

2 1 2

4



I dag

I Vi blar om til kapittel 3 i SICP.

I Tilstand og verditilordning.

I Destruktive operasjoner.

I Prosedyrebasert objektorientering + innkapsling.

I Ny modell for evaluering: omgivelsesmodellen.

Identitet og endring

I I kapittel 3 i SICP tar vi et stort skritt:

I Bryter med ren funksjonell programmering og introduserer elementer
fra imperativ programmering:

I Verditilordning som modifiserer objekter.

I Fra et funksjonelt perspektiv: destruktive operasjoner.

I Når vi endrer tilstanden til et objekt “destruerer” vi
det siden det ikke lenger representerer samme verdi.

I Likhet ved identitet vs verdi.

I Til nå har vi kunnet tenke på variabler som navn på verdier.

I Fra nå må vi tenke på variabler som steder man kan lagre verdier.

Tid og tilstand

I Så langt: et statisk univers.

I Har ikke forholdt oss til tid.

I Sentralt konsept i SICP kap. 3: tilstand.

I Hva mener vi med at noe har tilstand?

I At atferden avhenger av egenskaper som kan
forandre seg i forhold til tid og historikk.

I Kap. 3 introduserer også (prosedyrebasert) objektorientering.

I Viktig teknikk for å organisere programmer: Vi kan opprette objekter i
programmene våre som tilsvarer entiteter i verden vi ønsker å modellere.

Eksempel på objekt med tilstand: bankkonto

I Klassisk eksempel: bankkonto.

I Saldoen = tilstandsvariabel som oppsummerer
historikken (transaksjoner over tid).

I Hva vi kan gjøre, f.eks uttak, avhenger av saldoen.

I Kan definere en variabel balance for å representere saldoen.

I Trenger nå en måte å endre verdien symbolet er bundet til: set!

? (define balance 100)

? (set! balance (- balance 20))

? balance → 80

set!

? (define balance 100)

? (set! balance (- balance 20))

? balance → 80

I set! er en special form som lar oss modifisere variabelbindinger.

I (set! 〈name〉 〈value〉)

I Endrer variabelen 〈name〉 til å få verdien av uttrykket 〈value〉.

I Brukes for sin effekt (tilstandsendringen), ikke returverdi.

I Uttales “sett bæng”.

I Konvensjon i Scheme å bruke ! ved operasjoner for tilstandsendring
med uspesifiert returverdi.

Bankkonto #1

(define balance 100)

(define withdraw
(lambda (amount)

(if (>= balance amount)
(begin

(set! balance (- balance amount))
balance)

"Insufficient funds")))

? (withdraw 25) → 75

? (withdraw 25) → 50

? (withdraw 60) → "Insufficient funds"

? (withdraw 15) → 35

? balance → 35

? (set! balance 1000)

I Hva er nytt?

I Et kall på withdraw har
bieffekter utover sin
returverdi: Verdien til
balance endret!

I Samme argumenter,
forskjellig returverdi!

I Hvorfor har vi aldri trengt
begin tidligere?

I Svakhet: Variabelen
balance er eksponert, kan
endres av kode utenfor
withdraw.

Bankkonto #1 og #2

Først, en ny versjon av #1
(define balance 100)
(define withdraw ;; cond i stedet for if+begin

(lambda (amount)
(cond ((>= balance amount)

(set! balance (- balance amount))
balance)

(else "Insufficient funds"))))

Bankkonto #2, med innkapslet saldo
(define withdraw

(let ((balance 100))
(lambda (amount)

(cond ((>= balance amount)
(set! balance (- balance amount))
balance)

(else "Insufficient funds")))))

Bankkonto #2 (forts.)

(define withdraw
(let ((balance 100))

(lambda (amount)
(cond ((>= balance amount)

(set! balance (- balance amount))
balance)
(else "Insufficient funds")))))

? (withdraw 25) → 75

I Gjemmer variabelen med en teknikk som kalles innkapsling i SICP.
I = closure i Lisp-sjargong.

I let etablerer en ny omgivelse med balance som lokal variabel.
I balance er nå privat: kun tilgang via withdraw.
I Merk: Via prosedyren som returneres får vi tilgang til balance utenfor

det som ellers ville være rekkevidden av dens binding (let-uttrykket)!
I Kombinasjonen set! og lokale variabler lar oss modellere lokal tilstand.

LOL

I Eksempel på et nyttig idiom;

I let-over-lambda

I Enkelt eksempel på bruk av
innkapsling + verditilordning.

(define counter
(let ((x 0))

(lambda ()
(set! x (+ x 1))
x)))

? (counter) → 1

? (counter) → 2

? (counter) → 3

13



Bankkonto #3

I Svakhet med withdraw:
Fortsatt kun én konto.

I make-withdraw: lager
nye kontoobjekter.

I Returnerer en anonym
prosedyre som innkapsler
sin egen saldovariabel:

I Gitt som parameter i
stedet for via let.

I w1 og w2 er distinkte
objekter, med distinkte
lokale tilstandsvariabler.

(define make-withdraw
(lambda (balance)

(lambda (amount)
(cond ((>= balance amount)

(set! balance (- balance amount))
balance)

(else "Insufficient funds")))))

? (make-withdraw 100) → #<procedure>

? (define w1 (make-withdraw 100))

? (define w2 (make-withdraw 200))

? (w1 50) → 50

? (w2 50) → 150

Objektorientering via innkapsling og set!

I Closure-oriented programming:
I I eksemplene våre har vi representerert kontoobjekter som prosedyrer.
I Kombinert med modifisering av innkapslete lokale variabler for å

modellere tilstand.
I Guy Steele om å designe Scheme:

One of the conclusions that we reached was that the “object” need not be
a primitive notion in a programming language; one can build objects and
their behaviour from little more than assignable value cells and good old
lambda expressions.

λ + !

set!

I Hva har skjedd med språket vårt
etter at vi innførte set!?

I Koden kan ha tilstandsvariabler
– lokale eller globale – der
bindingen kan endre seg.

I En prosedyre som avhenger av
tilstand kan gi ulike returverdier
for like argumenter.

I Kan også modifisere tilstandsvariabler: vi sier da at prosedyren har
sideeffekter (tidligere trengte vi kun forholde oss til returverdi).

I Uttrykk kan evalueres i sekvenser der rekkefølge har blitt viktig.

I Må tenke på variabler som steder for å lagre verdier (ikke bare navn).

Repetisjon: Substitusjonsmodellen

I Standardregelen for evaluering av uttrykk
gjelder også for egendefinerte prosedyrer:

I Først evalueres deluttrykkene, så kalles
prosedyren på argumentverdiene.

I I substitusjonsmodellen tenker vi at
prosedyreuttrykk omskrives slik:
I Hele uttrykket erstattes av prosedyre-

kroppen fra definisjonen.
I Parameternavnene erstattes av

argumentverdiene.

Eksempel
? (define (square x)

(* x x))

? (square (+ 1 2))

⇒ (square 3)

⇒ (* 3 3)

→ 9

argumentverdiene.

Eksempel
? (define (square x)

(* x x))

? (square (+ 1 2))

⇒ (square 3)

⇒ (* 3 3)

→ 9

17

Substitusjonsmodellen holder ikke lenger

I La oss forsøke å bruke
substitusjonsmodellen her.

I Først et par ting å huske på:

- Første argumentet til set!
evalueres ikke (variabelen
som skal tilordnes verdi).

- Ved en sekvens av uttrykk
er det verdien av det siste
som returneres.

I Variablen balance brukes
tre ganger i kroppen:
Substitueres 2. og 3. bruk
samtidig blir resultatet feil.

? (define (make-withdraw balance)
(lambda (amount)

(set! balance (- balance amount))
balance))

? ((make-withdraw 25) 20)
→ 5

? ((make-withdraw 25) 20)

⇒ ((lambda (amount)
(set! balance (- 25 amount))
25)

20)

⇒ (set! balance (- 25 20))
25

⇒ (set! balance 5)
25

→ 25

En ny modell for evaluering

I Vi skal etablere en ny modell for hvordan kode evalueres.

I Omgivelsesmodellen.

I Sagt litt enkelt er det denne modellen vi skal implementere når vi lager
vår egen Scheme-evaluator i kapittel 4.

19



Omgivelser, rammer og bindinger

I En omgivelse er en sekvens av rammer.

I En ramme er en tabell som binder variabelnavn til verdier.

I Hver ramme har en peker til sin omsluttende omgivelse.

I Den globale omgivelsen består av én enkelt ramme
(med bl.a. bindingene for primitive prosedyrer).

I Variabler har sin verdi relativ til
en gitt omgivelse, gitt ved første
rammen med en binding.

I Nøstete bindinger for samme
navn overskygger hverandre.

Omgivelser og rammer: prosedyrer

I define legger til bindinger i en ramme, set! endrer dem.

I En prosedyre lages ved å evaluere et lambda-uttrykk i en gitt
omgivelse, og representeres ved et “par” som består av:

I Parametere og prosedyrekropp.

I En peker til omgivelsen der den ble opprettet.

? (define (square x)
(* x x))

Omgivelser og rammer: prosedyrekall

I Når en prosedyre anvendes, utvides prosedyrens omgivelse med en
ny ramme som binder parametrene til argumentverdiene:

I Prosedyrekroppen evalueres så i denne nye omgivelsen.

? (square 5)

Omgivelsesmodellen for evaluering

Evalueringsregel for sammensatte uttrykk
I Evaluer deluttrykkene rekursivt.
I Anvend operatorverdien på operandverdiene.

I For å anvende en prosedyre på argumenter,
I utvid omgivelsen til prosedyren med en ny ramme der de formelle

parametrene er bundet til argumentene den kalles på,
I og evaluer prosedyrekroppen i denne nye omgivelsen.

I Fortsetter helt til uttrykkene er redusert til atomære uttrykk og
primitive prosedyrer (kalles direkte uten ny omgivelse).

I Special forms (syntaktiske makroer) har egne evalueringsregler.

Tilbake til innkapsling

I Innkapsling kan nå avmystifiseres på bakgrunn av omgivelsesmodellen:

? (define make-withdraw
(lambda (balance)

(lambda (amount)
(if (>= balance amount)

(begin (set! balance (- balance amount))
balance)

"insufficient funds"))))

? (define w1 (make-withdraw 100))

? (w1 50) → 50

I Et kall på make-withdraw etablerer en ny ramme, som så blir
omgivelsen til den anonyme prosedyren den returnerer via lambda.

Tilbake til innkapsling (forts.)

I Resultatet av å definere make-withdraw i den globale omgivelsen.

Tilbake til innkapsling (forts.)

? (define w1 (make-withdraw 100))

Tilbake til innkapsling (forts.)

? (w1 50) → 50

Tilbake til innkapsling (forts.)

I Etter at kallet på w1 er ferdig utført.

Tilbake til innkapsling (forts.)

? (define w2 (make-withdraw 100))

Prosedyrebasert objektorientering og message passing

(define (make-account balance)
(define (deposit amount)

(set! balance (+ balance amount))
balance)

(define (withdraw amount)
(set! balance (- balance amount))
balance)

(define (dispatch message)
(cond ((eq? message ’deposit) deposit)

((eq? message ’withdraw) withdraw)
((eq? message ’balance) balance)))

dispatch)

;; Litt syntaktisk sukker:
(define (deposit amount account)

((account ’deposit) amount))

(define (withdraw amount account)
((account ’withdraw) amount))

(define (balance account)
(account ’balance))

Brukseksempler

? (define a1
(make-account 100))

? (define a2
(make-account 0))

? ((a1 ’deposit) 200)
→ 300

? (a2 ’balance) → 0

;; nytt grensesnitt:
? (deposit 200 a1) → 500

? (balance a2) → 0

? (eq? a1 a2) → #f

? (define a3 a2)

? (eq? a3 a2) → #t

? (deposit 42 a3) → 42

? (balance a2) → 42

Likhet (Identitet vs. ekvivalens)

I Typespesifikke predikater; = og
string=?

I equal? tester strukturell
ekvivalens rekursivt.

I eq? gir #t hvis argumentene
angir samme objekt i minnet.

I Merk at symboler og den
tomme lista '() er konstanter.

? (define bim '(1 2 (3)))

? (define bam '(1 2 (3)))

? (equal? bim bam) → #t

? (eq? bim bam) → #f

? (define bom bam) ;; aliasing

? (eq? bom bam) → #t

? (eq? '(1) '(1)) → #f

? (eq? '() '()) → #t

? (eq? 'foo 'foo) → #t

? (eq? + +) → #t

? (eq? 2.5 2.5) → ???

? (eq? "foo" "foo") → ???

